ATS

Jules Ferry

TP-info 2 : Utilisation du solveur d'équation différentielle intégré à Scilab

Info

<u>**Compétences nécessaires associées**</u> **:**

- **définir une fonction ;**
- **tracer un graphe d'une fonction** ;
- **créer un tableau à 2 lignes.**

I. Compétences nécessaires associées

1. Fonction

La définition d'un fonction f se fait de la manière suivante :

```
function y=f(x)    //définition de la fonction f de variable x et d'image y
    y=(x^2+2*x)*exp(-x)    //on définit  f(x)=(x²+2x)e^-x
endfunction
```

Les variables x et y utilisées dans la définition d'une fonction sont des variables muettes, elles peuvent donc être réutilisées dans la définition d'une nouvelle fonction ou dans Scilab.

2. Tracer un graphe d'une fonction

Si on veut tracer un graphe d'une fonction, on fait appel à la commande plot, il suffit d'avoir définit au préalable la liste x des variables dont on veut tracer la fonction, la commande plot calculera automatiquement la liste des images y associées :

```
function y=f(x)    //définition de la fonction f de variable x et d'image y
    y=(x^2+2*x)*exp(-x)
endfunction

x=linspace(-2,5,50);    //on veut tracer de x=-2 à x=5 avec 50 points

clf()    //fermer les fenêtres
plot(x,f)    //tracer le graphique
```

3. Superposer deux graphes

Si deux fonctions f et g ont été définies au préalable, on peut alors les superposer sur un même graphe par :

```
x=linspace(-2,5,50);    //on veut tracer de x=-2 à x=5 avec 50 points

clf()    //fermer les fenêtres
plot(x,f,x,g)    //tracer le graphique
```

II. La commande ode

1. Introduction

La commande *ode* est un solveur d'équation différentielle interne à Scilab. Elle fait appel à différents algorithmes connus pour résoudre les équations différentielles mais choisit la plus adaptées à la forme de l'équation différentielle utilisée. Il s'agit bien sûr d'une approximation.
Par exemple, pour les équations différentielles linéaires d'ordre un, la commande ode applique une méthode d'Euler. Si l'équation différentielle à résoudre est un oscillateur harmonique, elle fera appel à un autre algorithme (comme une méthode de Runge-Kutta d'ordre 2 ou 4 par exemple).

2. Utilisation sur l'exemple de la désintégration radioactive (premier ordre)

Prenons l'exemple de la désintégration radioactive : on considère un échantillon initial de N_0 noyaux radioactifs à $t=0$. On admet que leur désintégration suit la loi classique : $\frac{dN}{dt}=-\lambda N(t)$ où λ est appelée constante de désintégration radioactive (s'exprime en s^{-1}).

<u>Remarque</u> : nous sommes dans un cas particulier où $\frac{dN}{dt}$ ne dépend pas explicitement de t (on parle d'équation autonome) mais il faudra quand même mettre t en variable explicite dans l'algorithme pour le bon fonctionnement de la commande ode.

La commande *ode* s'utilise comme une fonction de 4 variables : `N = ode(N0,t0,t,Nprime)` où les 2 premières variables représentent les conditions initiales, la troisième variable représente la liste des instants t auxquels on veut connaître la liste des valeurs N associées et *Nprime* est la fonction (à définir au préalable) qui définit l'équation différentielle.

Exemple d'algorithme :

```
clear;   //effacer toutes les variables en mémoire

N0 = 1000; //définition très simple des conditions initiales
t0 = 0;
lambda = 1; //constante de désintégration radioactive en s-1

t = linspace(0,10,100); //on veut résoudre pour t entre 0 et 10s avec 100 points

function f = Nprime(t, N) //définition de la fonction N' qui va définir l'ED

f = -lambda*N

endfunction

N = ode(N0,t0,t,Nprime) //calcul de la liste N aux différents instants t

clf()
plot(t,N)
xlabel("t(s)")
ylabel("N")
title("N(t)")
```

3. Utilisation sur l'exemple de l'oscillateur harmonique (second ordre)

L'utilisation de la commande *ode* est inchangée pour une équation différentielle du second ordre. Toute la difficulté réside dans la fonction qui va définir l'équation différentielle !

Prenons l'exemple sur un oscillateur harmonique de position à l'équilibre nulle : $\ddot{x}+\omega_0^2 x(t)=0$ soit $\ddot{x}=-\omega_0^2 x(t)$.
L'idée ici est de définir l'équation différentielle non plus sur une fonction classique (ordre un) mais sur une fonction vectorielle à deux composantes (une double fonction). En reliant les deux composantes de manière adaptée, on pourra définir l'équation du second ordre comme si on avait une équation différentielle du premier ordre mais sur une fonction vectorielle :
on définit la fonction vectorielle $X'=(\dot{x};\ddot{x})$ à partir de $X=(x;\dot{x})$ de la manière suivante : $\begin{cases} X'_1=X_2 \\ X'_2=-\omega_0^2 X_1 \end{cases}$
(la première ligne est du bon sens, la seconde définit l'équation différentielle).

Exemple d'algorithme :

```
clear;

w0 = 2;    //pulsation propre

function f=Xprime(t, X)

f(1)=X(2)    //1ère composante de X'= 2ème composante de X

f(2)=-w0^2*X(1)    // 2ème composante de X' = -w0^2.1ère composante de X

endfunction

X0 = [1;0]; // C.I. rentrées comme un vecteur
//alternative : x0=1; v0=0; X0=[x0;v0]

t0 = 0;

t = linspace(0,10,1000);

X = ode(X0,t0,t,Xprime);

clf();
plot(t,X(1,:))// on ne trace que la 1ère ligne de X, soit x(t) et pas v(t)


plot(t,X(2,:))// si on veut tracer aussi v(t) : 2ème ligne de X
plot(t,X(1,:),t,X(2,:)) // variante si on veut superposer directement x(t) et v(t) de 2
couleurs différentes
```

III.Exploitation (TD)

1. Oscillateur amorti

Tracer $x(t)$ pour un oscillateur amorti de pulsation propre $\omega_0 = 2\, rad.s^{-1}$, de facteur de qualité $Q = 3$ et de position à l'équilibre $x_{eq} = 0$.

Remarque : une fois l'algorithme réalisé, amusez-vous à changer la valeur de Q et des conditions initiales.

2. Portrait de phase

Réaliser une trajectoire de phase de l'oscillateur amorti précédent.

3. Le pendule simple non amorti

On rappelle que l'équation différentielle qui régit l'évolution temporelle du pendule est non linéaire : $\ddot{\theta} + \frac{g}{l}\sin(\theta) = 0$.

1. Superposer deux courbes $\theta(t)$ pour des conditions initiales différentes mais restant dans le cadre de l'approximation de l'oscillateur harmonique.
   Que peut-on dire des périodes ?
2. Superposer deux courbes $\theta(t)$ pour des conditions initiales différentes dont l'une reste dans le cadre de l'approximation de l'oscillateur harmonique et l'autre est extrême ($\theta(0) = 179°$ et $\dot{\theta}(0) = 0$ par exemple).
3. Réaliser un portrait de phase (superposition de plusieurs trajectoires de phase) du pendule simple.